

## Lesson 14

### The Gradient Vector, Directional Derivatives

#### ▼ Initializations

```
> restart;  
with(VectorCalculus):  
BasisFormat(false):  
>
```

#### ▼ 14.1 The Gradient vector

Maple's **Gradient** routine is located in the **Vector Calculus** package. The syntax is

$$\mathbf{Gradient}(\text{Expression}, \mathbf{L})$$

where  $\mathbf{L}$  denotes a list of variables.

Technically the **Gradient** routine returns a vector field, not a free vector. This is important when non-Cartesian coordinates are used. Since we will need the gradient only in Cartesian coordinates, we will convert our results to a free vector using the **convert(vector field, Vector)** command.

#### ▼ Examples

##### ▼ Example 14.1.1

Compute the gradient of  $f(x, y) = x^2 y^3$ .

##### Solution

Code the function and use the **Gradient** routine.

```
> f:=(x, y)->x^2*y^3;  
gf:=Gradient(f(x,y), [x,y]):  
gf:=convert(gf, Vector);
```

$$f := (x, y) \rightarrow x^2 y^3$$

$$gf := \begin{bmatrix} 2xy^3 \\ 3x^2y^2 \end{bmatrix}$$

(2.1.1.1)

```
>
```

#### ▼ 14.2 Directional Derivatives

## Examples

### Example 14.2.1

Find the directional derivative of  $f(x, y) = \sin(2x) + e^{xy}$  in the direction of  $\mathbf{v} = 5\mathbf{i} - 12\mathbf{j}$ .

#### Solution

First compute  $\nabla f$  and the unit vector in the direction of  $\mathbf{v}$ .

```
> f:=(x, y)->sin(2*x)+exp(x*y);  
gf:=Gradient(f(x,y), [x,y]):  
gf:=convert(gf, Vector);
```

$$f := (x, y) \rightarrow \sin(2x) + e^{xy}$$

$$gf := \begin{bmatrix} 2 \cos(2x) + y e^{xy} \\ x e^{xy} \end{bmatrix} \quad (3.1.1.1)$$

```
> v:=<5, -12>;  
u:=v/Norm(v);
```

$$v := \begin{bmatrix} 5 \\ -12 \end{bmatrix}$$

$$u := \begin{bmatrix} \frac{5}{13} \\ -\frac{12}{13} \end{bmatrix} \quad (3.1.1.2)$$

The directional derivative of  $f$  in the direction of the vector  $\mathbf{v}$  is the dot product  $\nabla f \cdot \mathbf{u}$ .

```
> Du_f:=gf.u;
```

$$Du_f := \frac{10}{13} \cos(2x) + \frac{5}{13} y e^{xy} - \frac{12}{13} x e^{xy} \quad (3.1.1.3)$$

```
>
```

### Example 14.2.2

Find the directional derivative of  $f(x, y, z) = x \arctan\left(\frac{y}{z}\right)$  in the direction of  $\mathbf{v} = \mathbf{i} + \mathbf{j} - \mathbf{k}$  at the point  $P = (1, 2, -2)$ .

#### Solution

Code the function  $f$  and compute  $\nabla f$ .

```
> f:=(x, y, z)->x*arctan(y/z);  
gf:=Gradient(f(x, y, z), [x, y, z]):  
gf:=convert(gf, Vector);
```

$$f := (x, y, z) \rightarrow x \arctan\left(y \frac{1}{z}\right)$$

$$gf := \begin{bmatrix} \arctan\left(\frac{y}{z}\right) \\ \frac{x}{z\left(1 + \frac{y^2}{z^2}\right)} \\ -\frac{xy}{z^2\left(1 + \frac{y^2}{z^2}\right)} \end{bmatrix} \quad (3.1.2.1)$$

`> gf:=simplify(gf);`

$$gf := \begin{bmatrix} \arctan\left(\frac{y}{z}\right) \\ \frac{xz}{z^2 + y^2} \\ -\frac{xy}{z^2 + y^2} \end{bmatrix} \quad (3.1.2.2)$$

Next, we evaluate  $\nabla f$  at the point  $P$ .

`> gf_P:=eval(gf, {x=1, y=2, z=-2});`

$$gf_P := \begin{bmatrix} -\frac{1}{4}\pi \\ -\frac{1}{4} \\ -\frac{1}{4} \end{bmatrix} \quad (3.1.2.3)$$

Finally, we create the unit vector in the direction of  $\mathbf{v}$  and finish the problem.

`> v:=<1, 1, -1>;`  
`u:=v/Norm(v);`

$$v := \begin{bmatrix} 1 \\ 1 \\ -1 \end{bmatrix}$$

$$u := \begin{bmatrix} \frac{1}{3}\sqrt{3} \\ \frac{1}{3}\sqrt{3} \\ -\frac{1}{3}\sqrt{3} \end{bmatrix} \quad (3.1.2.4)$$

`> answer:=gf_P.u;`

$$\text{answer} := -\frac{1}{12} \pi \sqrt{3} \quad (3.1.2.5)$$

**Example 13.2.3.**

Compute an equation of the tangent plane and the normal line at the point  $P = (3, 2, 1)$  to the level surface

$$x y + x z + y z^3 = 11$$

**Solution**

The normal vector to the tangent plane is given by the gradient of the left hand side of the surface equation.

```
> f:=(x, y, z)->x*y+x*z+y*z^3;
gf:=Gradient(f(x, y, z), [x, y, z]);
gf:=convert(gf, Vector);
```

$$f := (x, y, z) \rightarrow x y + x z + y z^3$$

$$gf := \begin{bmatrix} y + z \\ x + z^3 \\ x + 3 y z^2 \end{bmatrix} \quad (3.1.3.1)$$

```
> n:=eval(gf, {x=3, y=2, z=1});
```

$$n := \begin{bmatrix} 3 \\ 4 \\ 9 \end{bmatrix} \quad (3.1.3.2)$$

The tangent plane is given by.

```
> TP:=n[1]*(x-3)+n[2]*(y-2)+n[3]*(z-1)=0;
TP := 3 x - 26 + 4 y + 9 z = 0 \quad (3.1.3.3)
```

```
> sort(TP, [x, y, z]);
3 x + 4 y + 9 z - 26 = 0 \quad (3.1.3.4)
```

To find the vector equation of the normal line we can use  $\mathbf{r}_0 = \langle 3, 2, 1 \rangle$  and  $\mathbf{v} = \mathbf{n} = \langle 3, 4, 9 \rangle$ .

```
> r0:=<3, 2, 1>;
```

$$r_0 := \begin{bmatrix} 3 \\ 2 \\ 1 \end{bmatrix} \quad (3.1.3.5)$$

The normal line is given by

```
> NL:=r0+t*n;
```

$$NL := \begin{bmatrix} 3 + 3 t \\ 2 + 4 t \\ 1 + 9 t \end{bmatrix} \quad (3.1.3.6)$$

L L LL >